Free Speech with a Lisp

6.001 Spring 1998

Lecture notes, February 26

PICS: Platform for Internet Content Selection

See also http://www.w3.org/PICS

PICS Statement of Principles

PICS is a cross-industry working group whose goal is to facilitate the development of technologies to give users of interactive media, such as the Internet, control over the kinds of material to which they and their children have access. PICS members believe that individuals, groups and businesses should have easy access to the widest possible range of content selection products, and a diversity of voluntary rating systems. 

In order to advance its goals, PICS will devise a set of standards that facilitate the following: 

Self-rating: 

enable content providers to voluntarily label the content they create and distribute. 

Third-party rating: 

enable multiple, independent labeling services to associate additional labels with content created and distributed by others. Services may devise their own labeling systems, and the same content may receive different labels from different services. 

Ease-of-use: 

enable parents and teachers to use ratings and labels from a diversity of sources to control the information that children under their supervision receive. 

PICS members believe that an open labeling platform which incorporates these features provides the best way to preserve and enhance the vibrancy and diversity of the Internet. Easy access to technology which enables first- and third-party rating of content will give users maximum control over the content they receive without requiring new restrictions on content providers. 

Membership in PICS includes a broad cross-section of companies from the computer, communications, and content industries, as well as trade associations and public interest groups. PICS member will deploy products and services based on these standards. 

PICS Label

(PICS-1.0  http://www.rsac.org/ratingsv01.html

     l gen true comment "RSACi North America Server"

     by rodgerb@playboy.com

     for http://www.playboy.com

     on "1996.04.04T08:15-0500"

     r (n 4 s 3 v 0 l 4))) 

PICS Service Description

 ((PICS-version 1.1)
  (rating-system "http://www.rsac.org/ratingsv01.html")
  (rating-service "http://www.rsac.org/")
  (name "The RSAC Ratings Service")
  (description "The Recreational Software Advisory Council rating

service.  Based on the work of Dr. Donald F. Roberts of Stanford

University, who has studied the effects of media on children for

nearly 20 years.")

  (default (label-only true))

  (category 

    (transmit-as "v")

    (name "Violence")
    (label (name "Conflict")

           (description "Harmless conflict; some damage to objects")

           (value 0))

    (label 

      (name "Fighting")

(description 

        "Creatures injured or killed; damage to objects; fighting")
      (value 1))

    (label

      (name "Killing")

(description

        "Humans injured or killed with small amount of blood")
      (value 2))
    (label

      (name "Blood and Gore")
      (description "Humans injured or killed; blood and gore")
      (value 3))

    (label

      (name "Wanton Violence")

(description "Wanton and gratuitous violence; torture; rape")
      (value 4)))

   . . .)

PICS Rules File

(PicsRule-1.1

 (Name (Rulename "user1.rlz"


  Description "user1.rlz created by RulesGenerator")


  Source (SourceURL "http://title.w3.org/"


  CreationTool "RulesGenerator/0.40"


  Author "jmiller@mit.edu"


  LastModified "1998-02-24T10:04-0500")

  Serviceinfo (Name http://www.rsac.org/
               Shortname "rsaci"


         BureauURL "http://www.rsac.org/ratingsv01.html"


         BureauUnavailable "FAIL")

  Policy (RejectUnless "((rsaci.l < 3) and (rsaci.n < 1))")

  Policy (AcceptIf "((rsaci.l = 0) and (rsaci.s = 0))")))

Symbols: A New Data Type

· Data types we have already seen: numbers, booleans, strings, procedures, pairs

· Each data type has a predicate to test if a value is of the given type: number? boolean? string? procedure? pair?

· All of Scheme’s built-in procedures require their arguments to be of specific types (see the Scheme manual).  For example, all of the arguments to + must be numbers.
· Today we introduce a new built-in procedure, eq? , which is Scheme’s fastest and most discerning test for equality.  Two things are eq? only if there is no way whatsoever  to tell them apart.  For now you only need to understand how it works on symbols: If two symbols look the same, they are the same.
Consider the PICS label above.  It looks like a list of 15 elements, some of which are strings (they have double quotes around them, like “1996.04.04T08:15-0500”) and one of which is another list.  But what are things like l, gen, true, and comment?  

Answer: a new data type, symbol.  The only thing special about symbols is how they are treated by eq?.  Are there any Scheme expressions whose values are symbols?  There aren’t any variables built in to Scheme whose values are symbols.  But there is a new special form that can be used to create symbols.

The value of ‘X is the symbol that prints out as X
Some examples:

(define A ‘X)

A                     =>  X

(eq? a ‘X)            => #T

(eq? ‘X ‘X)           => #T

(define B (list a 2))

B                     => (X 2)

(eq? (car b) ‘X)      => #T

One final note.  As a convenience, the single quote can be used to create data structures as well as symbols.   Examples:

(define C ‘(X 2))

C                     => (X 2)

(pair? C)             => #T

(eq? (car c) ‘X)      => #T

(define d ‘(A 1 B))

(car d)               => A

(length d)            => 3

(list-ref d 2)        => B

Processing Labels and Service Descriptions

Here is a program, expand, which takes as input a PICS label and a PICS Service Description and produces as output the rating from the label turned into English by inserting information from the Service Description to replace the “transmit name” and numeric value that appear in the label.

For example, using the label and service description from above:

(expand test-label test-service)

Produces:

(("Violence" #f

    (#f "Harmless conflict; some damage to objects"))

 ("Sex" #f (#f "Non-explicit sexual activity"))

 ("Nudity" #f (#f "Provocative frontal nudity"))

 (#f "Language" 

    (#f "Crude or explicit sexual references")))

This is the sort of program you might use to display a label for a document before showing the document itself, to allow a user to decide whether they are interested in viewing the document.  In fact, you might want to use PICS labels and PICSRules to set up criteria that specify things like:

· What labels on documents indicate that you are willing to just look at them automatically

· What labels on documents indicate that you’d like to see the label and approve before seeing the document itself

· What labels on documents indicate that you would rather not see the document unless you override your preferences.

One of the fascinating ideas being explored (both in research and in products around the Web) is to use PICS to specify preferences – so that a PICSRules file might be combined with some kind of push technology to personalize the information you receive.  Another  idea is to use some sort of community of users to automatically generate PICS labels or to locate documents in which you might be interested.  Libraries are exploring using a future version of PICS (called RDF) to create the equivalent of library card catalog records to help people search for information.

This code is available on-line at the 6.001 Web Site

(define (assq name a-list 
  ;; A-LIST has the form ((<name> ...) (<name> ...) ...)

  ;; Returns either (<name> ...) where <name> is EQ? to name

  ;;         or #F if no such entry is found

  (cond ((null? a-list) #F)

        ((eq? name (car (car a-list)))

         (car a-list))

        (else (assq name (cdr a-list)))))

(define (alist->value name a-list)

  (let ((result (assq name a-list)))

    (if result (second result) #F)))

(define (category->name category)

  (alist->value 'NAME (cdr category)))

(define (category->description category)

  (alist->value 'DESCRIPTION (cdr category)))

(define (category->transmit-name category)

  (string->symbol (alist->value 'TRANSMIT-AS (cdr category))))

(define (label->name label)

  (alist->value 'NAME (cdr label)))

(define (label->description label)

  (alist->value 'DESCRIPTION (cdr label)))

(define (rating->value rating)

  (second rating))

(define (rating->name rating)

  (first rating))

(define (category? entry) (eq? (car entry) 'CATEGORY))

(define (label? entry) (eq? (car entry) 'LABEL))

(define (value? entry) (eq? (car entry) 'VALUE))

(define (the l)

  (cond ((null? l) '())

        ((null? (cdr l))

         (first l))

        (else (error "Not unique" l)))) 

 (define (expand picslabel service)

  (let ((ratings (picslabel->rating picslabel)))

    (map (lambda (category)

           (get-name-and-value category ratings))

         (filter category? service))))

(define (get-name-and-value category ratings)

  (list (category->name category)

        (category->description category)

        (value->name category

         (the (filter (lambda (rating)

                        (eq? (category->transmit-name category)

                             (rating->name rating)))

                      ratings)))))

 (define (value->name category rating)

  (if (null? rating)

      ‘(NO VALUE FOR THIS CATEGORY)

      (let ((value (rating->value rating)))

        (let ((label

               (the (filter

                     (lambda (label)

                       (let ((values (filter value? label)))

                         (if (null? values)

                             #F

                             (= value (second (first values))))))

                     (map cdr (filter label? (cdr category)))))))

          (if label

              (list (label->name label) (label->description label))

              '(VALUE HAS NO NAME))))))

 (define (picslabel->rating label)

  (define (find-any-word words l)

    (cond ((null? l) #F)

          ((memq (car l) words) l)

          (else (find-any-word words (cdr l)))))

  (define (fix plist)

    (if (null? plist)

        '()

        (cons (list (first plist) (second plist))

              (fix (cddr plist)))))

  (let ((after-rating-keyword

         (find-any-word '(r ratings) label)))

    (if after-rating-keyword

        (if (cdr after-rating-keyword)

            (fix (second after-rating-keyword))

            #F))))
;; (expand test-label test-service)

;; ;Value: 

;; (("Violence" #f (#f "Harmless conflict; some damage to objects"))

;;  ("Sex" #f (#f "Non-explicit sexual activity"))

;;  ("Nudity" #f (#f "Provocative frontal nudity"))

;;  (#f "Language" (#f "Crude or explicit sexual references")))

(define (low-nudity pics-label)

  (let ((ratings (picslabel->rating pics-label)))

    (let ((nudity (assq 'n ratings)))

      (and nudity (< (second nudity) 3)))))

(define (low-violence pics-label)

  (let ((ratings (picslabel->rating pics-label)))

    (let ((violence (assq 'v ratings)))

      (and violence (< (second violence) 1)))))

;; (low-nudity test-label) => #F

;; (low-violence test-label) => #T

